

## 第 12 课 星期查询——字符串函数、键盘输入数据的函数

### 任务导航

本课将在设计“星期早知道”和“中小学生体育达标人数统计”的程序过程中，体会数组在程序中的应用，学习通过键盘输入数据的方法和字符串函数的应用。

编个程序输入明年的某月某日，输出该日是星期几。



图 12.1 “星期早知道”程序运行的界面

### Follow Me

#### 1. 星期早知道

##### 任务分析:

如果想知道明年的某月某日是星期几，只要记住明年 12 个月的代码就可以将明年任意一天是星期几算出来。

计算的方法：(月代码+日期)除以 7 的余数，即为该日期的星期数（若余数为 0 结果是星期日）。其中，月代码为上个月最后一天的星期数。

步骤 1：根据图 12.1 设计窗体界面。

##### DIY

试着根据图 12.1 所示的程序界面设计对象属性。

步骤 2：查找 2013 年的月代码

双击屏幕右下角的系统时钟，在出现的“日期和时间 属性”界面中查找 2013 年各月的代码（也可以在年历片中查找各月的月代码）。

我们查到 2013 年各月的代码按顺序为：144025036146

步骤 3：为查询准备数据（程序的初始化事件）

在初始化事件中，首先将存放代码的字符串每个数字字符进行分离，分离出的字符存放于月代码数组 a 的相应下标的下标变量中，以备使用。

对于计算出的星期数，要转换为该数字对应的汉字字符。例如：0 转换为“日”、1 转换为“一”等。

```
Dim a(1 To 12) As Integer '定义存放月代码的数组
Dim week(6) As String '定义存放汉字日期字符的数组
Dim dm As String '定义存放代码字符串的变量
'数据准备
Private Sub Form_Load()
    dm = "144025036146" '2013 年月代码字符串
    For i = 1 To 12
        a(i) = Val(Mid(dm, i, 1)) '将月代码存储到代码数组中
    Next i
    '星期标识转换
    week(0) = "日": week(1) = "一"
```

```

week(2) = "二": week(3) = "三"
week(4) = "四": week(5) = "五"
week(6) = "六"

```

End Sub

智多星

VB 中提供了子字符串函数和数字字符转换为数值型数据的函数，见表 12-1。

表 12-1 子字符串函数和字符转换为数值的函数

函 数	功 能	举 例	结 果
Mid(d, m, n)	从 d 字符串中第 m 个字符开始取 n 个字符	Mid("87654321", 3, 2)	"65"
Lift(d, n)	从 d 字符串中左侧向左开始取 n 个字符	Lift("87654321", 3)	"876"
Right(d, n)	从 d 字符串中右侧开始向右取 n 个字符	Right("87654321", 3)	"321"
Val(d)	将字符串 d 的数字字符转换为数值	Val("123")	123

在程序中我们使用了“子字符串函数”Mid(dm, i, 1)，其意义在于把字符串变量 dm 中的第 i 个位置开始的 1 个字符分离出。语句 a(i) = Val(Mid(dm, i, 1)) 将分离出来的这个字符转换为数值存放在 a(i) 中，也就是说 a(i) 中存放的是第 i 个月份的月代码。

步骤 4: 给标识为“查询结果”的命令按钮编写单击事件的过程代码

单击“查询结果”按钮后程序要将 Text1 中输入的要查询的月份数和 Text2 中输入的要查询的日期数，分别存放在变量 m 和 d 中，而 m 月的月代码是存放在 a(m) 中的，这样根据“星期早知道”的公式就容易求出 2013 某月某日是星期几了。

根据分析写出的程序代码如下：

```

Rem 查询星期
Private Sub Command1_Click()
    Dim m As Integer, d As Integer, t As Integer
    If Text1.Text <> "" And Text2.Text <> "" Then ' 文本框非空
        Command1.SetFocus ' 焦点定位在标识为“查询结果”的命令按钮上
        m = Text1.Text ' 输入的月份
        d = Text2.Text ' 输入的日期
        t = (a(m) + d) Mod 7 ' 计算星期
        Label3.Caption = "星期" & week(t) ' 显示结果
    Else
        Text1.SetFocus ' 文本框的输入为空时焦点定位在文本框 1 上
    End If
End Sub

```

步骤 5: 给标识为“清除”的命令按钮编写单击事件的过程代码

当单击“清除”按钮时，要求清除文本框中输入的内容和显示查询结果的标签框中的内容。

```

Rem 清除
Private Sub Command2_Click()
    Text1.Text = "" ' 文本框清空
    Text2.Text = ""
    Label3.Caption = "" ' 标签框清空
    Text1.SetFocus ' 焦点定位在文本框 1 上

```

End Sub

步骤 6: 保存、运行并调试程序

以“星期早知道.frm”为文件名保存文件于“第 3 单元\12 星期查询\星期早知道”的文件夹中。

2. 已知 2012 年的最后一天是星期一，编程序求 2013 年 12 个月的月代码。

任务分析:

因为已知 2012 年最后一天是星期一，所以相当于已知 2013 年 1 月份的月代码为 1。根据一月份的天数和月代码，利用公式  $(1 \text{ 月份代码} + 1 \text{ 月份天数}) \text{ Mod } 7$ ，就可以求出二月份的月代码，用同样的方法可以求出其他月份的月代码。

步骤 1: 编写求月代码的程序

```
Rem 求月代码
Dim a%(12)           ' %为整型变量定义符
Private Sub Form_Load()
    a(1) = 1         ' 保存 1 月份的月代码
    For m = 1 To 11
        Select Case m      ' 求 m 月的天数
            Case 2
                n = 28
            Case 4, 6, 9, 11
                n = 30
            Case 1, 3, 5, 7, 8, 10, 12
                n = 31
        End Select
        k = m + 1         ' 月代码在数组中的存放位置
        a(k) = (a(m) + n) Mod 7 ' 计算并保存第 k 月的月代码
    Next m
End Sub
Rem 显示月代码
Private Sub Form_Activate()
    For k = 1 To 12
        Print a(k);
    Next k
End Sub
```

步骤 2: 保存、运行并调试程序

以“求月代码.frm”为文件名保存文件于“第 3 单元\12 星期查询\求月代码”的文件夹中。

智多星

如果知道 2012 年最后一天是星期一，即可通过程序求出 2013 年 12 个月的月代码，那么“2013 年星期早知道”就可以不用在年历上去查询 12 个月的月代码了。

DIY

假设我们只知道 2012 年的最后一天是星期一和 2013 年各月份的天数，试编程序求 2013 年的某月某日是星期几。

3. 中小学生体育达标人数统计

对中小学生有体育达标的规定，达标活动规定跑、跳、投项目都通过基本要求的同学为达标，为了统计出达标的人数，同学们设计了以下的程序，在程序运行时只要分别输入每项运动通过同学的编号，

程序就可以给出达标同学的编号。

表 12-2 中的数据分别给出了跑、跳、投通过同学的编号：

表 12-2 各项达标同学的编号

项目	项目通过同学的编号
跑	1, 7, 8, 6, 2, 4
跳	4, 8, 2, 3, 6
投	2, 6, 8, 5, 4

步骤 1：设计用户界面，如图 12.2。




图 12.2 达标统计的窗体界面

步骤 2：“达标统计”的程序代码

```

Dim s(1 To 8) As Integer
Private Sub Command1_Click()
    Picture1.Cls
    For i = 1 To 24
        x = Val(InputBox("请输入项目通过同学的编号", "数据输入"))
        If x = -1 Then      ' 设-1 为数据输入的退出标志
            Exit For      ' 退出循环
        Else
            s(x) = s(x) + 1  ' 编号输入次数的统计
            If s(x) = 3 Then ' 下标为编号的下标变量值是 3 表示该编号的同学达标
                Picture1.Print Space(2) & "编号为" & CStr(x) & "号的同学达标"
            End If
        End If
    Next i
End Sub

```

智多星：

在程序运行过程中可以用 InputBox ( ) 函数为变量提供数据，程序执行函数时将按照函数后面括号内的参数制作如图 12.3 所示的对话框提示用户输入数据。




图 12.3 InputBox ( ) 函数产生的对话框

### 键盘输入函数

格式:


<变量>=InputBox(“输入提示字符串”“标题栏字符串”,<默认输入值>)

功能: 在程序的运行过程中通过对话框为变量提供数据

用 InputBox 函数为变量提供的数据默认为字符型数据。若要获得的数据为数值型，可以用 Val 函数将其转换为数值型数据，也可以将变量定义为数值型后再使用 InputBox 函数为变量提供数据。

### DIY

参考下面给出的输出 8 月份月历的程序代码，编写 2013 年各月份月历浏览的程序。程序界面如下图 12.4 所示。



(a) 程序运行前的月历浏览界面

(b) 程序运行后的月历浏览界面

图 12.4 2013 年月历窗体界面

### 智多星

组合框 (ComboBox)

功能: 既能像列表框一样从已有的项中进行选择，又能自己输入选项。

常用方法和属性

组合框的 AddItem 方法、RemoveItem 方法，ListIndex 属性、ListCount 属性等的使用和列表框相同。

Style 属性: 当值为 0-Dropdown 时，可以在组合框的文本框部分输入信息，也可以在列表部分选择项目。当值为 2-Dropdown 时，只能在列表中选择项目。

显示 8 月份月历

```
Private Sub Command1_Click()  
    k = 4                '保存月历第一天的星期  
    Picture1.Cls  
    Picture1.Print  
    For i = 0 To 6      '输出月历表头  
        Picture1.Print Tab(6 * i + 4); '打印语句可使用 Tab 函数控制数据的输出位置  
        Picture1.Print week(i);  
    Next i  
    For i = 1 To 31    '输出月历日期  
        Picture1.Print Tab(6 * k + 3); '日期输出定位  
        Picture1.Print i;  
        If k = 6 Then '输出月历时的换行控制  
            k = 0: Picture1.Print  
        Else  
            k = k + 1  
        End If  
    Next i  
End Sub
```

步骤 3: 保存、运行并调试程序

以“达标统计.frm”为名保存程序到“第 3 单元\12 星期查询\达标统计”的文件夹中。

知识着陆

1. 数组主要用于批量有序数据的处理。
2. 使用子字符串函数可以将一个字符串的一部分分离出来形成一个新的字符串。
3. VB 提供了在程序运行过程中为变量提供数据的函数 InputBox(“数据输入”), 该函数允许用户在程序的运行过程中为变量提供数据。

视野拓展

VB 中的函数

函数是用来完成特定功能的独立程序代码, 函数可以返回一个值给调用程序。

在 VB 中可以调用系统函数, 也可以调用自己定义的函数。

系统函数是系统给出的具有一定功能的程序代码。例如: LoadPicture() 就是一个系统函数, 它的功能是载入一张图片; sin(a)、cos(a) 也是系统函数, 它们的功能是计算角的正弦值和余弦值; Rnd 随即函数也是系统函数, 它的功能是产生一个 0 到 1 的随机数。

创作天地

1. 参考如图 12.5 的窗体设计, 产生 100 个 10 以内的随机整数, 统计每个数出现的次数。




图 12.5

2. 参考如图 12.6 的窗体设计, 将 0 到 10 的汉语拼音存储于 a 数组的 a(0)到 a(10)中, 输入 10 以内的整数, 输出该数的汉语拼音。




图 12.6

3. 参考如图 12.7 的窗体设计, 输入百以内的整数, 输出这个百以内整数的汉语拼音。



图 12.7

方法提示:

- 输入一个百以内的整数 x;
- 对 x 分离数字;  
n: 存放十位数字; m: 存放个位数字
- 根据读音分析设计程序  
读音分析:

如果  $n=0$  则  
    输出一个汉语拼音  
否则 如果  $m=0$  则  
    输出两个汉语拼音  
    否则  
        输出三个汉语拼音  
    判断处理结束  
判断处理结束